



教育部大学计算机课程改革项目规划教材

Visual Basic程序设计 实验教程

陈宏明 章 慧 周 蕾 主编

TP312BA-33
18



高等教育出版社
HIGHER EDUCATION PRESS

014032273

TP312BA-33
18



教育部大学计算机课程改革项目规划教材

Visual Basic 程序设计 实 验 教 程

Visual Basic Chengxu Sheji Shiyan Jiaocheng

陈宏明 章 慧 周 蕾 主编



高等教育出版社·北京
HIGHER EDUCATION PRESS BEIJING



北航

C1720521

TP312BA-33
18

内容提要

本书是主教材《Visual Basic 程序设计教程》(周蕾、王留洋, 高等教育出版社)的配套实验指导书。书中共安排了 16 个实验, 内容涉及 VB 集成开发环境、菜单及多窗体、顺序结构、分支结构、循环结构、数组、过程和图形处理等。每个实验按照教学的重点内容, 由浅入深、循序渐进地安排了多个实验项目, 注重培养学生的应用能力。

本书的使用对象主要为应用型本科院校非计算机专业学生, 也可供计算机专业学生和广大编程爱好者自学或参考。

图书在版编目(CIP)数据

Visual Basic 程序设计实验教程 / 陈宏明, 章慧,
周蕾主编. —北京: 高等教育出版社, 2014.1

ISBN 978-7-04-039285-2

I. ①V… II. ①陈… ②章… ③周… III. ①BASIC 语言—程序设计—高等学校—教材 IV. ①TP312

中国版本图书馆 CIP 数据核字 (2014) 第 007352 号

策划编辑 唐德凯
版式设计 范晓红

责任编辑 唐德凯
插图绘制 黄建英

特约编辑 刘亚军
责任校对 刘娟娟

封面设计 于文燕
责任印制 张泽业

出版发行 高等教育出版社
社 址 北京市西城区德外大街 4 号
邮政编码 100120
印 刷 北京市四季青双青印刷厂
开 本 787mm × 1092mm 1/16
印 张 7.25
字 数 170 千字
购书热线 010-58581118

咨询电话 400-810-0598
网 址 <http://www.hep.edu.cn>
<http://www.hep.com.cn>
网上订购 <http://www.landaco.com>
<http://www.landaco.com.cn>
版 次 2014 年 1 月第 1 版
印 次 2014 年 1 月第 1 次印刷
定 价 13.80 元

本书如有缺页、倒页、脱页等质量问题, 请到所购图书销售部门联系调换

版权所有 侵权必究

物 料 号 39285-00

前言

“Visual Basic 程序设计”是一门具有较强理论性和实践性的公共基础课。本书作为《Visual Basic 程序设计教程》(周蕾、王留洋,高等教育出版社)的配套实验指导书,在编写过程中充分考虑了应用型本科非计算机专业学生的特点,帮助学生在实验中把握重点,突破难点,掌握面向对象程序设计的基本思路和方法。

本书充分体现了教学内容组织的科学性和灵活性,将理论教学与实践教学相融合,在理论中引导实践,在实践中体现理论,循序渐进地引导学生学习程序设计的基本方法及技巧。

本书共安排了 16 个实验,主要包括 VB 集成开发环境、界面设计、程序流程控制结构、数组、过程、文件和图形处理等内容。每个实验分为实验目的、相关知识和实验内容三个部分。实验目的部分指明了实验的目的和意义;相关知识部分给出了实验内容所涉及的理论基础知识;实验内容部分由若干相对独立的实验项目构成,每个实验项目都配有实验要求、实验分析和实验步骤等,以帮助学生理解实验内容,提高编程技巧。

本书由陈宏明、章慧、周蕾主编,王红华、李芬芬、唐朝霞、王留洋参与了编写,全书由陈宏明、周蕾统稿。

在本书的编写过程中参考了有关教材、手册和资料,并得到了校、院领导和众多教师的支持和帮助,在此向他们表示真诚的感谢。

由于编者水平有限,书中难免有错误和不足之处,敬请广大读者批评指正,编者的 E-mail 地址: memory@hyit.edu.cn。

编 者

目录

实验一	VB 集成开发环境的使用	1
实验二	界面设计	8
实验三	菜单及多窗体程序设计	16
实验四	顺序结构程序设计	21
实验五	分支结构程序设计	30
实验六	循环结构程序设计	36
实验七	循环嵌套及常用算法程序设计	42
实验八	固定大小数组程序设计	49
实验九	动态数组及数组常用算法程序设计	58
实验十	控件数组	68
实验十一	其他控件的使用	74
实验十二	通用过程程序设计	80
实验十三	参数传递与递归过程	86
实验十四	文件	92
实验十五	文件管理控件和通用对话框控件的使用	100
实验十六	图形处理	105
参考文献		110

实验一 VB 集成开发环境的使用

一、实验目的

- (1) 掌握 VB 6.0 的启动与退出。
- (2) 了解 VB 6.0 的集成开发环境，熟悉各主要窗口的作用。
- (3) 了解 VB 6.0 应用程序的开发过程，熟悉 VB 的三种工作模式。
- (4) 理解 VB 中对象的概念。
- (5) 掌握窗体、标签、命令按钮和文本框的基本使用方法。
- (6) 了解用 VB 制作简单动画的过程。

二、相关知识

1. VB 6.0 的集成开发环境

VB 6.0 的集成开发环境由菜单、工具栏、工具箱及各种特定功能的窗口组成。

2. 创建 VB 应用程序的步骤

创建 VB 应用程序的一般步骤：

- (1) 设计用户界面。
- (2) 设置对象属性。
- (3) 编写代码。
- (4) 调试和运行程序。
- (5) 保存程序。

3. VB 的工作模式

VB 有三种工作模式，即设计模式、运行模式和中断模式。

4. VB 中的对象及其要素

VB 中对象是指应用程序中具有特殊属性的基本实体。对象三要素包括属性、方法和事件。

5. VB 中对象的公共属性和方法

VB 中很多对象都有一些公共的属性和方法。常用的公共属性有 Name、Caption、Enabled、Visible 等；常用的公共方法有 Move 等。

6. VB 中制作动画的关键控件——计时器控件

在 VB 中，制作动画的一个关键控件是计时器控件。这个控件一般用来检查系统时钟，判断是否应该执行某项任务。

三、实验内容

实验 1.1 启动 VB 6.0，创建一个“标准 EXE”工程。熟悉集成开发环境中的各个组

成部分及三种工作模式，并能熟练地在各个窗口间切换。

【实验步骤】

(1) 启动 VB，在“新建工程”对话框中选择“标准 EXE”选项，单击“打开”按钮，进入 VB 集成开发环境，如图 1.1 所示。标题栏中的标题栏显示[设计]，表明集成开发环境处于设计模式。




图 1.1 VB 集成开发环境

(2) 单击工具栏中的“启动”按钮或按 F5 键，标题栏中的标题显示“运行”，表明集成开发环境处于运行模式。

(3) 在“运行”菜单中选择“中断”命令，标题栏中的标题显示“break”，表明集成开发环境处于中断模式。

(4) 单击工具栏上的“结束”按钮，可停止程序运行，回到设计模式，单击“继续”按钮可继续运行程序，进入运行模式。

(5) 在集成开发环境中分别寻找“窗体设计器”窗口、“属性”窗口、“工程资源管理器”窗口、“窗体布局”窗口、“工具箱”窗口等，熟悉它们的默认位置。

(6) 双击“窗体设计器”窗口，进入“代码”窗口。在“视图”菜单中切换显示“代码”窗口和“窗体”窗口。

(7) 在“视图”菜单中单击“立即”窗口，在“立即”窗口中输入 `print 1+2`，按 Enter 键，观察运行结果；再输入 `print 2>1`，按 Enter 键，观察运行结果。结果如图 1.2 所示。

(8) 分别关闭“属性”窗口、“工程资源管理器”窗口、“工具箱”窗口，然后再用“视图”菜单中对应的菜单命令将其显示。

实验 1.2 在标题为“改变窗体背景色”窗体上添加两个命令按钮，其名称分别为 C1、C2，标题分别为“改变颜色”和“退出”。要求：程序运行时，窗体的标题栏中没有任何控制按钮；单击命令按钮 C1，将窗体的背景颜色改为红色 (&HFF&)，如图 1.3 所示；单

击命令按钮 C2, 结束程序运行。将窗体文件保存为 EX1-2.frm, 工程文件保存为 EX1-2.vbp; 对保存过的 VB 工程编译并生成可执行文件 EX1-2.exe。

【分析】窗体的 ControlBox 属性用于设置是否在窗体上显示控制菜单栏, 默认值为 True, 本实验将其设置为 False 即可。




图 1.2 “立即”窗口




图 1.3 实验 1.2 的运行界面

【实验步骤】

(1) 建立一个新的工程。在 VB 集成开发环境中单击“文件”菜单, 选择“新建工程”选项。

(2) 选中窗体, 在属性窗口设置窗体属性 (见表 1.1)。

表 1.1 窗体属性设置

对 象	属 性	设 置 值
窗体 1	Caption	改变窗体背景色
	ControlBox	False

(3) 在窗体上添加两个命令按钮, 设置控件属性 (见表 1.2)。

表 1.2 控件属性设置

对 象	属 性	设 置 值
命令按钮 1	Name	C1
	Caption	改变颜色
命令按钮 2	Name	C2
	Caption	退出

(4) 添加程序代码:

```
Private Sub C1_Click()  
    Form1.BackColor = vbRed  
End Sub  
Private Sub C2_Click()  
    End  
End Sub
```

(5) 运行程序。单击工具栏中的“启动”按钮或选择“运行”→“启动”菜单选项。

(6) 保存文件。选择“文件”→“保存工程”菜单项, 首先弹出“文件另存为”对话

框，选择保存位置，将窗体文件保存为 EX1-2.frm；然后弹出“工程另存为”对话框，将工程文件保存为 EX1-2.vbp。

(7) 生成可执行文件。选择“文件”→“生成 EX1-2.exe”菜单项，弹出“生成工程”对话框，此时可执行文件名已默认为 EX1-2.exe，如图 1.4 所示。选择保存位置，单击“确定”按钮。

(8) 运行可执行文件。双击 EX1-2.exe 文件，运行该可执行文件。

注意：新生成的可执行文件无需进入 VB 环境，在 Windows 中能够直接运行。

实验 1.3 在名称为 F1，标题为“简单 VB 实验”的窗体上添加一个标签控件，其名称为 Lbshow，标题为“VB 程序设计语言”；再添加两个命令按钮，名称分别为 Cmdshow 和 Cmdclear，标题分别为“显示”和“隐藏”，如图 1.5 所示。要求：单击“隐藏”按钮，窗体上的文字消失；单击“显示”按钮，窗体上的文字又恢复显示。

【分析】 通过设置标签的 Visible 属性值可以实现程序运行期间标签控件是否可见。




图 1.4 “生成工程”对话框




图 1.5 实验 1.3 的运行界面

【实验步骤】

- (1) 按图 1.5 所示设计程序界面。
- (2) 属性设置（见表 1.3）。

表 1.3 控件属性设置

对 象	属 性	设 置 值
窗体 1	Name	F1
	Caption	简单 VB 实验
标签 1	Name	Lbshow
	Caption	VB 程序设计语言
命令按钮 1	Name	Cmdshow
	Caption	显示
命令按钮 2	Name	Cmdclear
	Caption	隐藏

- (3) 添加程序代码：

```
Private Sub Cmdshow_Click()
```

```

Lbshow.Visible = True
End Sub
Private Sub Cmdclear_Click()
    Lbshow.Visible = False
End Sub

```

(4) 将窗体文件保存为 EX1-3.frm, 工程文件保存为 EX1-3.vbp, 运行工程并验证结果。

实验 1.4 在窗体上添加一个有边框的标签控件, 其标题为“欢迎学习 VB!”; 添加两个命令按钮, 其标题分别为“左移”和“右移”; 设置窗体的高和宽分别为 3000 和 4000, 窗体的标题栏中没有最大化和最小化按钮, 如图 1.6 所示。要求: 单击“左移”按钮, 标签向左移动 100 特维; 单击“右移”按钮, 标签向右移动 100 特维; 单击窗体, 则标签移到窗体的右上角, 如图 1.7 所示。




图 1.6 实验 1.4 的运行界面




图 1.7 单击窗体时的运行界面

【分析】 移动标签可执行 Move 方法, 也可通过设置控件的 Left 属性 (左右移动) 或 Top 属性 (上下移动) 实现。Move 方法的语法为

[object.]Move Left [,Top] [,Width] [,Height]

省略对象名时表示移动的是当前窗体。

【实验步骤】

- (1) 按图 1.6 所示设计程序界面。
- (2) 属性设置 (见表 1.4)。

表 1.4 控件属性设置

对 象	属 性	设 置 值
窗体 1	Height	3000
	Width	4000
	MaxButton	False
	MinButton	False
标签 1	Caption	欢迎学习 VB!
	BorderStyle	1
命令按钮 1	Caption	左移
命令按钮 2	Caption	右移

- (3) 完善程序代码:

```

Private Sub Command1_Click()
    Label1.Move Label1.Left - 100 '实现标签的左移

```

```
End Sub

Private Sub Command2_Click()
    Label1.Left= Label1.Left+100    '实现标签的右移
End Sub

Private Sub Form_Click()
    Label1.Top = 0                    '利用 Top 和 Left 属性设置对象位置
    Label1.Left = Form1.Width - Label1.Width
End Sub
```

(4) 将窗体文件保存为 EX1-4.frm，工程文件保存为 EX1-4.vbp，运行工程并验证结果。

实验 1.5 在名称为 Form1 的窗体上画一个图片框控件和两个标题为“开始”和“停止”的命令按钮，再在窗体上放置一个定时器控件。利用“属性”窗口，加载一个图片到图片框控件，界面如图 1.8 所示。要求：单击“开始”按钮，图片框控件在屏幕上每隔 0.5 秒钟向下移动 100 特维，当图片框控件移出屏幕下方后，会从屏幕上方出现继续下移；单击“停止”按钮，图片框控件停止移动。



图 1.8 实验 1.5 的程序界面

【分析】 图片框每隔 0.5 秒自动下移，需要借助定时器控件实现。要使定时器控件有效，需要具备两个条件：Enabled 属性设置为 True；Interval 属性设置为大于 0 的值。定时器的 Interval 属性用来设置定时器自动触发 Timer 事件的时间间隔，单位为毫秒，本实验定时器的 Interval 属性应设置为 500。

【实验步骤】

- (1) 按图 1.8 所示设计程序界面。
- (2) 属性设置（见表 1.5）。

表 1.5 控件属性设置

对 象	属 性	设 置 值
图片框 1	Picture	单击... 装入图形
	Autosize	True
计时器 1	Enabled	False
	Interval	500
命令按钮 1	Caption	开始
命令按钮 2	Caption	停止

(3) 完善程序并调试运行。程序代码如下：

```
Private Sub Command1_Click()
    Timer1.Enabled = True
End Sub
```

```

Private Sub Command2_Click()
    Timer1.Enabled = False
End Sub

Private Sub Timer1_Timer()
    Picture1.Move Picture1.Left, Picture1.Top + 100
    If Picture1.Top > Form1.Height Then
        Picture1.Top = - Picture1.Height
    End If
End Sub

```

(4) 将窗体文件保存为 EX1-5.frm, 工程文件保存为 EX1-5.vbp, 运行工程并验证结果。

实验二 界面设计

一、实验目的

- (1) 掌握窗体、标签和文本框常用属性、方法和事件。
- (2) 掌握命令按钮控件的常用属性、方法和事件。
- (3) 理解框架控件的作用。
- (4) 掌握单选按钮、复选框控件的常用属性、方法和事件。
- (5) 掌握滚动条控件的常用属性和事件。
- (6) 掌握计时器的常用属性与事件。
- (7) 掌握图片框、图像框控件的常用属性。

二、相关知识

1. 窗体

窗体是所有对象的容器，具有自己的属性、方法和事件。在程序设计阶段，窗体是程序员的工作台，而在运行阶段，每个窗体对应一个窗口。

2. 标签与文本框

与文本有关的标准控件有两个，即标签和文本框。在标签中只能显示文本，不能进行编辑，而在文本框中既可以显示文本，又可以编辑文本。

3. 命令按钮

命令按钮通常用来在单击时执行指定的操作。

4. 框架

框架的作用是将控件分组，以便把不同种类的控件分隔开。框架在实际运用中往往和其他控件一起使用。

5. 单选按钮与复选框

两者均可用来表示选择状态，其中单选按钮是互斥的，在同一组中只能有一个单选按钮处于选中状态；复选框可以同时选中多个，也可以一个也不选。

6. 滚动条

滚动条通常附在窗口上用来帮助用户观察数据或确定位置，也可作为数据输入的工具。

7. 定时器

定时器可以每隔一定时间间隔自动触发 **Timer** 事件，一般利用定时器控件完成简单的动画设计。

8. 图片框与图像框

图片框与图像框是两个用来显示图片的控件，除了显示图片外，图片框控件还可以作为其他控件的容器和输出的对象。

三、实验内容

实验 2.1 在窗体上添加两个标签，标题分别为“用户名”和“口令”；添加两个命令按钮，标题分别为“确定”和“退出”；再添加两个文本框，初始内容为空，如图 2.1 所示。要求：程序运行后，用户在两个文本框中分别输入用户名和口令，其中“口令”文本框中的输入信息以“*”号显示。单击“确定”按钮或按下 Enter 键，如果用户输入的姓名与密码相同，窗体上显示“登录成功”提示信息，如图 2.2 所示；否则，窗体上显示“登录失败”；单击“退出”按钮或按下 Esc 键，结束程序运行。

【分析】 利用文本框的 PasswordChar 属性，可将其设置为一个密码文本框，运行时在文本框中输入任意字符，都将显示为 PasswordChar 属性所设置的字符；设置命令按钮的 Default 属性为 True，则按下 Enter 键相当于单击该命令按钮；设置命令按钮的 Cancel 属性为 True，则按下 Esc 键相当于单击该命令按钮。




图 2.1 实验 2.1 的程序界面




图 2.2 实验 2.1 的运行界面

【实验步骤】

- (1) 按图 2.1 所示设计程序界面。
- (2) 属性设置（见表 2.1）。

表 2.1 属 性 设 置

对 象	属 性	设 置 值
窗体 1	Caption	登录系统
	BorderStyle	1
标签 1	Caption	用户名
标签 2	Caption	口令
文本框 1	Text	
文本框 2	PasswordChar	*
	Text	
命令按钮 1	Caption	确定
	Default	True
命令按钮 2	Caption	退出
	Cancel	True

(3) 添加程序代码:

```
Private Sub Command1_Click()  
    If Text1.Text = Text2.Text Then '判断两个文本框中输入内容是否相同  
        Print "登录成功"  
    Else  
        Print "登录失败"  
    End If  
End Sub  
  
Private Sub Command2_Click()  
    End '终止程序运行  
End Sub
```

(4) 将窗体文件保存为 EX2-1.frm，工程文件保存为 EX2-1.vbp，运行工程并验证结果。
注意：运行时请检验按 Enter 键和 Esc 键的执行结果。

实验 2.2 在窗体上放四个命令按钮和两个文本框控件，如图 2.3 所示。其中文本框 Text1 的高和宽分别为 1500 和 3000，可以显示多行文本，具有垂直滚动条。要求：运行程序，在文本框 Text1 中输入文本。选中部分文本内容，单击“复制”按钮，可将 Text1 中选中的文本内容复制到文本框 Text2 中；单击“剪切”按钮，可将 Text1 中选中的文本内容移到文本框 Text2 中；单击“清空”按钮，将文本框中的内容全部清空，文本框 Text1 获得焦点；单击“退出”按钮，结束程序运行。




图 2.3 实验 2.2 的运行界面

【分析】 文本框 Text 属性用于设置或返回文本框的文本内容，SelText 属性返回文本框运行时选中的文本内容。本实验可利用这两个属性实现文本框的赋值和选中内容的获取。

【实验步骤】

- (1) 按图 2.3 所示设计程序界面。
- (2) 属性设置（见表 2.2）。

表 2.2 属 性 设 置

对 象	属 性	设 置 值
窗体 1	Caption	简单文本编辑
文本框 1	Height	1500
	Width	3000
	MultiLine	True
	ScrollBars	2
	Text	
文本框 2	Text	

续表

对 象	属 性	设 置 值
命令按钮 1	Name	CmdCopy
	Caption	复制
命令按钮 2	Name	CmdCut
	Caption	剪切
命令按钮 3	Name	CmdClear
	Caption	清空
命令按钮 4	Name	CmdExit
	Caption	退出

(3) 完善程序代码:

```
Private Sub CmdCopy_Click() '复制
    Text2.Text = Text1.SelText
End Sub

Private Sub CmdCut_Click() '剪切
    Text2.Text = Text1.SelText
    Text1.SelText = ""
End Sub

Private Sub CmdClear_Click() '清空
    Text1.Text = ""
    Text2.Text = ""
    _____ '文本框 Text1 获得焦点
End Sub

Private Sub CmdExit_Click() '退出
    End
End Sub
```

(4) 将窗体文件保存为 EX2-2.frm，工程文件保存为 EX2-2.vbp，运行工程并验证结果。

实验 2.3 在窗体上添加两个框架，其标题分别为“字体”和“修饰”；再添加一个文本框，设置内容为“VB 实验指导书”，居中显示；在框架 1 中添加三个标题分别为“黑体”、“楷体”和“宋体”的单选按钮，在框架 2 中添加三个标题分别为“粗体”、“斜体”、“下划线”的复选框，如图 2.4 所示。要求：通过“属性”窗口设置相关属性，使得在初始状态下，“宋体”单选按钮处于选中状态，“下划线”的复选框处于选中状态，但以灰色显




图 2.4 实验 2.3 的运行界面

示。运行时选中单选按钮和复选框，则对文本框中的文字作相应的设置。

【分析】 单选按钮和复选框的选中状态均由 Value 属性决定，其中单选按钮 Value 属性值为布尔型，True 表示选中，False（默认值）表示未选中；复选框的 Value 属性值为数值型，0（默认值）表示未选中，1 表示选中，2 表示灰色，此时复选框呈现“√”标志，但以灰色显示，表示暂时不可用。

【实验步骤】

(1) 按图 2.4 所示设计程序界面。

注意：将控件放入框架控件时，应首先在窗体上放上框架控件，然后将其他控件添加在框架中，这样移动框架时，可以同时移动所包含的控件，保证框架是这些控件的容器。

(2) 属性设置（见表 2.3）。

表 2.3 属 性 设 置

对 象	属 性	设 置 值
框架 1	Caption	字体
框架 2	Caption	修饰
单选按钮 1	Caption	黑体
	Value	False
单选按钮 2	Caption	楷体
	Value	False
单选按钮 3	Caption	宋体
	Value	True
复选框 1	Caption	粗体
	Value	0
复选框 2	Caption	斜体
	Value	0
复选框 3	Caption	下划线
	Value	2
文本框 1	Text	VB 实验指导书
	Alignment	2-Center

(3) 程序代码：

```
Private Sub Option1_Click()  
    Text1.FontName = "黑体"  
End Sub  
  
Private Sub Option2_Click()  
    Text1.FontName = "楷体_GB2312"  
End Sub  
  
Private Sub Option3_Click()  
    Text1.FontName = "宋体"  
End Sub
```